************************************************
***Title: main.do
***Creators: Joelle Abramowitz, Shooshan Danagoulian, and Owen Fleming*
***Notes: This file produces the main estimates of the effect of pollen exposure on suicides. The preferred specification includes indicators for location-season specific quartiles of pollen, weather controls, and county, year-month, and month-day fixed effects.

*For questions, contact
*Owen Fleming
*hg3490@wayne.edu
************************************************


**********SETUP
use data/for_analysis, clear


**********PRODUCE ESTIMATES
*Generate sample
ppmlhdfe count pollen_q2_ls pollen_q3_ls pollen_q4_ls $weather, absorb(county year_month month_day) cluster(county) tolerance(1e-06)
gen sample = e(sample)

*Treatment: location-season quartiles
eststo main_s1: poisson count pollen_q2_ls pollen_q3_ls pollen_q4_ls $weather if sample==1, cluster(county) tolerance(1e-06)
estadd ysumm
estadd scalar counties = e(N_clust)
estadd local Controls "Weather" 
estadd local FE ""

*Treatment: location-season quartiles
eststo main_s2: ppmlhdfe count pollen_q2_ls pollen_q3_ls pollen_q4_ls $weather if sample==1, absorb(county month year) cluster(county) tolerance(1e-06)
estadd ysumm
estadd scalar counties = e(N_clust)
estadd local Controls "Weather" 
estadd local FE "County, Month, Year"

*Treatment: location-season quartiles
eststo main_s3: ppmlhdfe count pollen_q2_ls pollen_q3_ls pollen_q4_ls $weather if sample==1, absorb(county year_month month_day) cluster(county) tolerance(1e-06)
estadd ysumm
estadd scalar counties = e(N_clust)
estadd local Controls "Weather" 
estadd local FE "County, Year x Month, Month x Day"


**********EXPORT
estout using results/main.xls, cells(b(star label(Coef.) fmt(4)) se(par(`"="("'`")""') label(Std. Err.) fmt(4))) stats(ymean N counties Controls FE) starlevels(* 0.1 ** 0.05 *** 0.01) keep(pollen_q2_ls pollen_q3_ls pollen_q4_ls) legend label replace 
eststo clear


